2.1 Linear Equations: Method of Integrating Factors

Maple Setup

Load the DEtools package to make DEplot available so as to graph direction fields. End with a colon to suppress printing the list of routines. We will also be using solve and eval in this problem.

```
> with(DEtools):
```

Problem 21

(a) Draw a direction field for the given differential equation. How do the solutions appear to behave as t becomes large? Does the behavior depend on the choice of the initial value a? Let a_0 be the value of a for which the transition from one type of behavior to another occurs. Estimate the value of a_0.

(b) Solve the initial value problem and find the critical value a_0 exactly.

(c) Describe the behavior of the solution corresponding to the initial value a_0.

21. $y' - 1/2\, y = 2\cos(t)$

Define the differential equation remembering that Maple requires us to use $y(t)$, not y. We'll use the *operator notation* D(y) for the derivative function, y' is entered as D(y)(t).

```
> ode := D(y)(t) - 1/2*y(t) = 2*cos(t);
```

$$ode := \mathrm{D}(y)(t) - \frac{1}{2}\, y(t) = 2\cos(t)$$

Let's go directly to graphing a direction field. Remember, the syntax for DEplot is

DEplot(*differential_equation, dependent_var, independent_var_range, dependent_var_range***)**

Take y in [-8, 8] and t in [0, 10] to start.

```
> DEplot(ode, y(t), t=0..10, y=-8..8);
```



From the direction field, it appears that for $|a|$ > about 1, the curves go off to +/- infinity quickly. We should investigate values near zero. Let's add integral curves with a values of -1/2, -1/4, 0, 1/4, and 1/2. We create a list of points using square brackets naming it **InitialValues**.

```
> InitialValues := [[0,-1], [0,-1/2], [0,0], [0,1/2], [0,1]];
```

$$InitialValues := \left[[0, -1], \left[0, -\frac{1}{2}\right], [0, 0], \left[0, \frac{1}{2}\right], [0, 1] \right]$$

Now add the integral curves (coloring them black) to the direction field.

```
> DEplot(ode, y(t), t=0..10, y=-8..8, InitialValues, linecolor=
  black);
```



There is a dramatic difference between the integral curves for *a* = -1/2 and *a* = -1. Let's look more closely there. Use *a* values of -1, -5/6, -2/3, and -1/2.

```
> InitialValues := [[0,-1], [0,-5/6], [0,-2/3], [0,-1/2]];
  DEplot(ode, y(t), t=0..10, y=-8..8, InitialValues, linecolor=
  black);
```

$$InitialValues := \left[[0, -1], \left[0, -\frac{5}{6}\right], \left[0, -\frac{2}{3}\right], \left[0, -\frac{1}{2}\right] \right]$$



Let's try one more graph. Set *a* to be 4 equally spaced points from -2/3 up to -1/2.

```
> InitialValues := [[0,-5/6], [0,-7/9], [0,-13/18], [0,-1/2]];
  DEplot(ode, y(t), t=0..10, y=-8..8, InitialValues, linecolor=
  black);
```

$$InitialValues := \left[\left[0, -\frac{5}{6}\right], \left[0, -\frac{7}{9}\right], \left[0, -\frac{13}{18}\right], \left[0, -\frac{1}{2}\right]\right]$$



We could continue this process, but, for now, let's say the critical *a* value is about 0.7.

Our next task is to determine the exact critical value of *a* by solving the first-order linear equation symbolically.

We know multiplying by the integrating factor $\mu = e^{-\frac{t}{2}}$ makes the left side become the derivative of $e^{-\frac{t}{2}} y$. Integrating the right side and multiplying by $e^{\frac{t}{2}}$ will give us a solution. Don't forget the constant of integration!

```
> Soln := exp(t/2)*(int(exp(-t/2)*rhs(ode), t) +C);
```

$$Soln := e^{\frac{1}{2}t}\left(-\frac{4}{5}e^{-\frac{1}{2}t}\cos(t) + \frac{8}{5}e^{-\frac{1}{2}t}\sin(t) + C\right)$$

Have Maple **expand** this expression, and then **simplify** using the option **exp** to combine exponentials.

```
> Soln := expand(Soln);
  Soln := simplify(Soln, exp);
```

$$Soln := -\frac{4}{5}e^{\frac{1}{2}t}e^{-\frac{1}{2}t}\cos(t) + \frac{8}{5}e^{\frac{1}{2}t}e^{-\frac{1}{2}t}\sin(t) + e^{\frac{1}{2}t}C$$

$$Soln := -\frac{4}{5}\cos(t) + \frac{8}{5}\sin(t) + e^{\frac{1}{2}t}C$$

To determine C, set $y(0) = a$ and **solve** for C.

```
> solve(eval(Soln, t=0) = a, {C});
```

$$\left\{C = a + \frac{4}{5}\right\}$$

Substitute back into **Soln.**

```
> Soln := subs(C=4/5+a, Soln);
```

$$Soln := -\frac{4}{5}\cos(t) + \frac{8}{5}\sin(t) + e^{\frac{1}{2}t}\left(a + \frac{4}{5}\right)$$

We can see the critical value of a, the value that eliminates the exponential growth term, is $a = -4/5$.

Our last task is to describe the behavior we see below.

```
> InitialValues := [[0,-1], [0,-4/5], [0,-3/5]];
  DEplot(ode, y(t), t=0..10, y=-8..8, InitialValues, linecolor=
  black);
```

$$InitialValues := \left[[0, -1], \left[0, -\frac{4}{5}\right], \left[0, -\frac{3}{5}\right]\right]$$



>

For values of *a* larger than -4/5, solutions go to +infinity; for values of *a* less then -4/5, integral curves go to -infinity. For the critical value *a* = -4/5, the solution oscillates and is purely periodic.